

# DRC 오류 데이터베이스를 사용하여 LVL 실행 결과 분석

## 소개

Expert 레이아웃 에디터는 LVL (레이아웃 대 레이아웃), DRC (설계 규칙 검사) 및 Lisa (인터페이스 애플리케이션 언어, Expert 특수 목적 언어) 기능을 제공합니다. Expert 라이선스로 이것을 수행할 수 있습니다.

본 문서에서 DRC 오류 탐색 툴을 이용하여, LVL 실행 결과를 분석하는 방법에 대해 설명합니다.

## 개요

LVL은 프로젝트에서 두 셀 간의 차이점을 비교하고 탐지합니다. 이는 특정 셀에 대한 오류처럼 오류 데이터베이스에 저장할 수 있습니다.

현재 버전은 LVL에 대해 명령줄을 지원할 수 없으므로, GUI를 이용한 수동 작업이 필요합니다.

LVL 실행 후, Lisa를 활용하여, 자동화된 작업으로 이용할 수 있습니다.

LVL은 오류 레이어를 하나의 레이어로 통합하거나 그대로 유지할 수 있습니다. 후자의 경우 Expert 라이선스만으로 오류 데이터베이스에서 출력할 수 없으므로, 통합 옵션을 선택해야 합니다.




그림 1. LVL 설정

## LVL 실행

LVL은 현재 라이브러리 또는 활성화된 참조 라이브러리에 있는 셀을 지정할 수 있습니다. GDSII 파일을 사용하는 경우, ELD 파일로 변환된 참조 라이브러리를 설정하거나 현재 디렉토리로 가져와야 합니다.

Expert 메뉴에서 이를 호출할 수 있습니다.

[Verification]

>> [Layout vs. Layout]

지오메트리 (병합)는 개별 도형을 병합하는 것을 의미합니다.

계층 비교는 모든 계층 수준의 지오메트리를 활용하는 것을 의미합니다.

LVL에는 몇 가지 논리 연산이 있으며, 두 셀의 차이점을 확인할 때 XOR 연산을 사용합니다.




그림 2. LVL 오류의 결과로 생성된 오류 데이터베이스의 예

## 탐지된 셀의 DRC 오류 데이터베이스

탐지한 LVL의 차이점은 임시 셀 데이터로 저장되므로, 새로운 셀 이름으로 저장해야 합니다.

준비한 Lisa 스크립트를 사용하여, 임시 지오메트리 데이터를 DRC 오류 데이터베이스로 이동할 수 있습니다. 이때 가능한 모든 DRC 오류 검사 기능을 사용하여, 레이아웃 간의 차이점을 검토할 수 있습니다.

탐지한 차이점은 DRC 오류처럼 확인할 수 있습니다. 또한 플래그 레이어가 "\_As\_Is\_"로 설정된 경우, 상이한 레이어 이름이 팝업 플래그에 설명되어 있습니다.

Syntax : lv\_after

XI > lv\_after [ /base = A|B /flag=value ]

XI 명령줄에서 lv\_after를 실행합니다. 하나 또는 두 개의 셀을 기본 셀로 지정하고, 드롭다운 목록에서 레이어 이름을 선택할 수 있는 대화상자가 나타납니다. 레이어가 없으면 명령이 취소됩니다.

## 결론

여기서 소개한 모든 기능은 Expert 라이선스만으로 사용할 수 있습니다. 10종류 이상의 레이어가 있을 경우, 단일 레이어로 통합해야 합니다.

LVL은 GUI 모드에서 실행하므로, 설계에 대규모의 통합이 있으면 LVL 성능이 저하됩니다.

출력 오류 데이터베이스의 성능은 차이점의 탐지 갯수에 따라 달라집니다.




그림 3. 설명 오류의 예

```

!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!
!   Output Error DataBase after LVL                               !
!       *Presented by SILVACO Japan Co., Ltd.                   !
!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!
define command/replace "lvl_after";
define action
parameter base
parameter flag
do begin

!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!
!!! make "drop-down-list" of two cells. !!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!
!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!
    cur_cell = get_edited_cell_name();
    cell_buf = buffer_create();
    buffer_insert_at_end(cell_buf, &cur_cell[10..-1]);
    loop begin
        if(cell_buf.at_end) then (leave loop);
        buf_move = buffer_match(cell_buf, "_xor_");
        if(buf_move EQL False) then (buffer_advance(cell_buf, 1));
        if(buf_move EQL True) then begin
            buffer_advance(cell_buf, -5);
            buffer_replace(cell_buf, cell_buf.index, (cell_buf.index+4), "%t");
            leave loop;
        end;
    end;
    seq_cell = {(&cell_buf)[1..cell_buf.index-1], (&cell_buf)[(cell_buf.index+1)..-1]};

!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!
!!! make "drop-down-list" of all layers. !!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!
!!!   (Except for the Derive Layer, Datatype=/0 & GDSnumber<0. !!!!!!!!!!!!!
!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!
    layers = get_layer_list();
    c=1;
    loop begin
        if(c GTR layers.size) then (leave loop);
        if((get_gds_datatype(&layers[c]) NEQ 0) OR (get_gds_number(&layers[c]) LSS 1))
            then(seq_remove_ith(layers, c)) else(c=c+1);
    end;
    layers_buf = buffer_create();
    buffer_insert_at_end(layers_buf, &layers);
    loop begin
        if(layers_buf.at_end) then (leave loop);
        buffer_scan(layers_buf, "%n");
        buffer_replace(layers_buf, layers_buf.index, layers_buf.index, "%t");
        buffer_advance(layers_buf, 1);
    end;

!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!
!!! Action. !!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!
!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!
    if(&base EQL "A") then (cell_name = seq_cell[1])

```

```

elseif(&base EQL "B") then (cell_name = seq_cell[2])
else (cell_name = "");
layer_name = "_As_Is_";
buffer_insert(layers_buf, 1, (&layer_name &"%t"));

if((&base EQL "") OR (flag EQL 0)) then begin
  t1={ &cell_buf , "Base Cell Name"};
  t2={ &layers_buf , "Flag Layer Name"};
  tt={t1,t2};
  tt1={" Option Mode "};
  vars1 = (form create (tt) (tt1));

  if(&vars1.type NEQ "Null") then begin      !!! Form
    cell_name = vars1[1];
    layer_name = vars1[2];
  end; !!!!!!! Form Cancel !!!!!!!
end;

cell status /rw;
select all;
tools cutbyvertex /all /selected /regionvertices=512 /wirevertices=256;
copy selection;
!!!! Version 4.4.X !!!!
all_shapes = (find objects (SEARCH_ANY_SHAPE) /selected /seq_output);
x_pos = (all_shapes[1].bbox).xpos;
y_pos = (all_shapes[1].bbox).ypos;
if(all_shapes.size GEQ 2) then begin
  l = 1;
  loop begin
    l = l + 1;
    if(l GTR all_shapes.size) then (leave loop);
    if((all_shapes[l].bbox).xpos LSS x_pos) then (x_pos = (all_shapes[l].bbox).xpos);
    if((all_shapes[l].bbox).ypos LSS y_pos) then (y_pos = (all_shapes[l].bbox).ypos);
  end;
end;
!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!
cell close;

if((cell_exists(cell_name) EQL True) AND ((layer_exists(layer_name) EQL True) OR (&layer_name EQL "_As_Is_"))) then begin      !!! Execution
  if(cell_exists("lvl_result") EQL True) then (cell delete "lvl_result");
  cell new "lvl_result";
  cell open "lvl_result";
  instance 0 0 /cell = (cell_name);
  if(&layer_name EQL "_As_Is_") then begin
    paste selection (x_pos) (y_pos);          !!! Version 4.4.X
!     paste selection;                        !!! Version 4.5.X
    all_layers = get_layer_list();
    m = 0;
    n = 0;
    check = "";
    loop begin
      m = m + 1;

```

```

    if(m GTR all_layers.size) then (leave loop);
    layer_count = get_layer_object_count( (get_edited_cell_name()), (all_
layers[m]));

    if(layer_count GTR 0 ) then begin
        n = n + 1;
        if(n EQL 10) then begin
            ret = message_box("There are more than 10 different kinds of layers.%n"
                &"Guardian-DRC license is required.%n%n"
                &"Proceed to execute ?", {});
            if(ret EQL false) then (leave loop);
        end;
        check = &check &"Copy: Layer=" &all_layers[m] &";";
    end;
    if(m EQL all_layers.size) then (drc command (check) /cell); ! /selection
end;
end
else begin
    paste layer (layer_name);
    merge selection;
    tools cutbyvertex /all /selected /regionvertices=512 /wirevertices=256;
    zoom /all;
    check = "Copy: Layer=" &layer_name &";";
    drc command (check) /cell; ! /selection
end;

end;
!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!

end; !!!!!!! define end !!!!!!!
define argument base /named /coerce_to=(String) /default="" /is_default;
define argument flag /named /coerce_to=(Integer) /default=0 /is_default;
complete command;

```